
*Does decentralisation turn minority parties into secessionists? 
*Insights from Eastern and Western Europe
*Edina Szöcsik and Christina Isabel Zuber, West European Politics
*Analysis

use "Secessionism.dta", clear

*summary statistics*
sum sec2 sec_shift gov_ter2011 difgov_ter  ps_ter2011 difps_ter  self_reg2010 coinciding electab lnelectab gov kin
tab sec2 
tab sec_shift  
tab coinciding 
tab gov 
tab kin

*correlations*
pwcorr sec2 ps_ter2011 gov_ter2011 self_reg2010 coinciding lnelectab gov kin, star(.05) 
pwcorr sec_shift difps_ter difgov_ter difself_reg, star(.05) 

*** TABLES ***

*Table 3. Results of logit regression models of ethnonational minority parties’ secessionism (maximum likelihood estimation)*
use "Secessionism.dta", clear
logit sec2 ps_ter2011 lnelectab gov kin
estimates store m1
logit sec2 gov_ter2011 lnelectab gov kin
estimates store m2     
logit sec2 self_reg2010 lnelectab gov kin
estimates store m3      
logit sec2 coinciding lnelectab gov kin
estimates store m4      
estout m1 m2 m3 m4, cells(b(star fmt(3)) se(par fmt(2))) starlevels(∞ 0.10 * 0.05 ** 0.01 *** 0.001) stats(N) legend collabels(, none) varlabels(_cons Constant) posthead("") prefoot("") postfoot("") varwidth(16) modelwidth(12) delimiter("")


*Table 4. Comparing the probability of secessionist Hungarian parties in Romania with Catalan parties in Spain
*Romania (M1)
use "Secessionism.dta", clear
logit sec2 ps_ter2011 lnelectab gov kin
prvalue, x(ps_ter2011=1.8689473 lnelectab=3.7148397 gov=0 kin=1) 
*Spain (M1)
prvalue, x(ps_ter2011=5.8565154 lnelectab=4.2185955 gov=0 kin=0) 
*Romania, if parliamentarian parties of Romania had the positions of Spanish parliamentarian parties (M2)
prvalue, x(ps_ter2011=5.8565154 lnelectab=3.7148397 gov=0 kin=1)


*Romania (M2)
use "Secessionism.dta", clear
logit sec2 gov_ter2011 lnelectab gov kin
prvalue, x(gov_ter2011=3.500557 lnelectab=3.7148397 gov=0 kin=1) 
*Spain (M2)
prvalue, x(gov_ter2011=6.8 lnelectab=4.2185955 gov=0 kin=0) 
*Romania, if governing parties of Romania had the positions of Spanish governing parties (M2)
prvalue, x(gov_ter2011=6.8 lnelectab=3.7148397 gov=0 kin=1) 


*Romania (M3)
use "Secessionism.dta", clear
logit sec2 self_reg2010 lnelectab gov kin
prvalue, x(self_reg2010=10 lnelectab=3.7148397 gov=0 kin=1) 
*Spain (M3)
prvalue, x(self_reg2010=20.926949 lnelectab=4.2185955 gov=0 kin=0) 
*Romania if the level of selfrule in Romania was equally high as in Spain (M3)
prvalue, x(self_reg2010=20.926949 lnelectab=3.7148397 gov=0 kin=1) 


*Romania (M4)
use "Secessionism.dta", clear
logit sec2 coinciding lnelectab gov kin
prvalue, x(coinciding=0 lnelectab=3.7148397 gov=0 kin=1) 
*Spain (M4)
prvalue, x(coinciding=1 lnelectab=4.2185955 gov=0 kin=0) 
*Romania, if Hungarians had a coinciding region like the Catalans  (M4)
prvalue, x(coinciding=1 lnelectab=3.7148397 gov=0 kin=1) 


*Table 5. Results of first differences logit regression models of ethnonational minority parties’ shift to secessionism (penalized maximum likelihood estimation)* 
*correcting for rare event of shift towards secessionism by using Penalized Maximum Likelihood Estimation proposed by Firth 
*https://www3.nd.edu/~rwilliam/stats3/RareEvents.pdf
*findit firthlogit
use "Secessionism.dta", clear
firthlogit sec_shift difps_ter 
estimates store m5 
firthlogit sec_shift difgov_ter 
estimates store m6	              
estout m5 m6, cells(b(star fmt(3)) se(par fmt(2))) starlevels(∞ 0.10 * 0.05 ** 0.01 *** 0.001) stats(N) legend collabels(, none) varlabels(_cons Constant) posthead("") prefoot("") postfoot("") varwidth(16) modelwidth(12) delimiter("")


***ROBUSTNESS TESTS (see Online Appendix F)
*Table F1. Two-level logit regressions 
use "Secessionism.dta", clear
xtmelogit sec2 ps_ter2011   || country_id: , var
estimates store m7
xtmelogit sec2 ps_ter2011 lnelectab gov kin || country_id: , var 
estimates store m8

xtmelogit sec2  gov_ter2011 || country_id: , var 
estimates store m9
xtmelogit sec2 gov_ter2011 lnelectab gov kin  || country_id: , var
estimates store m10

xtmelogit sec2 self_reg2010 || country_id: , var     
estimates store m11 
xtmelogit sec2 self_reg2010 lnelectab gov kin || country_id: , var 
estimates store m12

xtmelogit sec2 coinciding  || country_id: , var
estimates store m13
xtmelogit sec2 coinciding lnelectab gov kin || country_id: , var 
estimates store m14
	   

*Table F2. Two-level models, first differences  
xtmelogit sec_shift difps_ter || country_id: , var 
estimates store m15
xtmelogit sec_shift difgov_ter || country_id: , var 
estimates store m16


*Table F3. Shift towards lower democratic quality in Turkey and Ukraine 
gen demworse = 0
replace demworse = 1 if country_name == "Turkey"
replace demworse = 1 if country_name == "Ukraine"

firthlogit sec_shift difps_ter demworse
estimates store m17
firthlogit sec_shift difgov_ter demworse
estimates store m18


*Table F4. excluding Bosnia 
use "Secessionism.dta", clear
drop if country_id == 79

logit sec2 ps_ter2011 lnelectab gov kin
estimates store m19
logit sec2 gov_ter2011 lnelectab gov kin
estimates store m20     
logit sec2 self_reg2010 lnelectab gov kin
estimates store m21      
logit sec2 coinciding lnelectab gov kin
estimates store m22      
estout m19 m20 m21 m22, cells(b(star fmt(3)) se(par fmt(2))) starlevels(∞ 0.10 * 0.05 ** 0.01 *** 0.001) stats(N) legend collabels(, none) varlabels(_cons Constant) posthead("") prefoot("") postfoot("") varwidth(16) modelwidth(12) delimiter("")


*Table F5. excluding Bosnia, first differences
use "Secessionism.dta", clear
drop if country_id == 79

firthlogit sec_shift difps_ter
estimates store m23 
firthlogit sec_shift difgov_ter
estimates store m24	              
estout m23 m24, cells(b(star fmt(3)) se(par fmt(2))) starlevels(∞ 0.10 * 0.05 ** 0.01 *** 0.001) stats(N) legend collabels(, none) varlabels(_cons Constant) posthead("") prefoot("") postfoot("") varwidth(16) modelwidth(12) delimiter("")



*** FIGURES ***

use "Secessionism.dta", clear

*Figures 1 and 2, box plots showing  accommodation in east and west 
bysort country_id: gen count = _n 
keep if count == 1
graph box gov_ter2011 ps_ter2011, over(east) mark(1, mlab(country_name)) mark(2, mlab(country_name)) mark(3, mlab(country_name))
*improve figure with graph editor
graph save Graph "Figure 1.gph"

graph box self_reg2010, over(east) mark(1, mlab(country_name)) mark(2, mlab(country_name)) mark(3, mlab(country_name))
*improve figure with graph editor
graph save Graph "Figure 2.gph"


*Displaying changes in programmatic accommodation
*Appendix C, Figure C1
use "Secessionism.dta", clear
graph bar (mean) difgov_ter difps_ter, over(country_name)
*improve figure with graph editor
graph save Graph "Figure Appendix C.gph"


